Example 4a: Multivariate General Linear Models for Repeated Measures in SAS and STATA
(complete syntax, data, and output available for SAS and STATA electronically)

These data were collected for my masters' thesis and are unpublished in this form (to see the way I'd prefer to have analyzed the data, see Hoffman & Rovine, 2007 *Behavior Research Methods* or chapter 12 of my textbook, *Longitudinal Analysis*). The outcome was the log-transformed mean per condition of response time to detect changes in driving scenes that were either of low/high meaningfulness to driving or low/high visual salience (i.e., a 2x2 repeated measures design). This sample includes 97 younger adults (age range = 18–32) and 59 older adults (age range = 63–86). We will specify piecewise linear effects of age to create a mean difference between younger and older adults and a linear age slope within the older adults. We will estimate multivariate models with normal conditional distributions using residual maximum likelihood (REML) and denominator degrees of freedom in SAS and STATA MIXED. Note that STATA provides incorrect AIC and BIC values using REML (it counts all parameters instead of variance parameters only), so those values are not referred to below.

Original data in wide format (was one row per person, outcomes in separate columns):

	PersonID: Person ID number	old: Is in Older Age Group 0=YA, 1=OA)	age: Actual Age in Years	rt11: Response Time (sec) for Low Meaning, Low Salience	rt12: Response Time (sec) for Low Meaning, High Salience	rt21: Response Time (sec) for High Meaning, Low Salience	rt22: Response Time (sec) for High Meaning, High Salience
97	112	0	27.00	12.410	5.524	10.114	7.435
98	201	1	77.00	15.087	10.099	15.957	13.502

New data in stacked format (one row per outcome per person) after transformation code below:

	PersonID: Person ID number	old: Is in Older Age Group 0=YA, 1=OA)	age: Actual Age in Years	condition: Index for Outcome (M/S)	mean: Meaning (0=Low, 1=High)	sal: Salience (0=Low, 1=High)	rt: Stacked Response Time across Conditions	logRT: Natural Log of Response Time	yrs65: Age in Older Adult Group (0=65)
385	112	0	27.00	11	0	0	12.410333333	2.5185294589	0
386	112	0	27.00	12	0	1	5.5239583333	1.7090946927	0
387	112	0	27.00	21	1	0	10.113680556	2.3138890178	0
388	112	0	27.00	22	1	1	7.435	2.0061985799	0
389	201	1	77.00	11	0	0	15.086736111	2.7138159546	12
390	201	1	77.00	12	0	1	10.098571429	2.3123939711	12
391	201	1	77.00	21	1	0	15.956517857	2.7698673888	12
392	201	1	77.00	22	1	1	13.502083333	2.6028439945	12

STATA Syntax for Importing and Stacking Wide into Univariate (now one row per outcome per person):

```
* Define global variable for file location to be replaced in code below
global filesave "C:\Dropbox\20_PSQF7375_Generalized\PSQF7375_Generalized_Example4a"

* Import example 4a multivariate data into work library
use "$filesave\Example4aWide.dta", clear

* Stack data: list multivariate variables first, i(personID) j(condition)
reshape long rt, i(personid) j(condition)

* Create condition variables
gen mean=0
gen sal=0
recode mean (0=1) if condition==21
recode mean (0=1) if condition==22
recode sal  (0=1) if condition==12
recode sal  (0=1) if condition==22
```

```
* Label new stacked variables
label variable condition "condition: Index for Outcome"
label variable mean "Meaning (0=Low, 1High)"
label variable sal "Salience (0=Low, 1=High)"
label variable rt "rt: Combined Response Time across Conditions"

* Create additional variables
gen logrt=ln(rt)
gen yrs65=0
replace yrs65=age-65 if old==1
* Label new variables
label variable logrt "logRT: Natural Log of Response Time"
label variable yrs65 "yrs65: Age in Older Adult Group (0=65)"
```

SAS Syntax for Importing and Stacking Wide into Univariate (now one row per outcome per person):

```
* Define global variable for file location to be replaced in code below;
%LET filesave= C:\Dropbox\20_PSQF7375_Generalized\PSQF7375_Generalized_Example4a;
* Location for SAS files for these models (uses macro variable filesave);
LIBNAME filesave "&filesave.";

* Import example 4a multivariate data into work library and stack it;
DATA work.Example4a; SET filesave.Example4aWide;
     condition=11; mean=0; sal=0; rt=rt11; OUTPUT; * Low  meaning, low  salience;
     condition=12; mean=0; sal=1; rt=rt12; OUTPUT; * Low  meaning, high salience;
     condition=21; mean=1; sal=0; rt=rt21; OUTPUT; * High meaning, low  salience;
     condition=22; mean=1; sal=1; rt=rt22; OUTPUT; * High meaning, high salience;
* Label new stacked variables;
  LABEL condition= "condition: Index for Outcome (M/S)"
        mean= "Meaning (0=Low, 1=High)"
        sal=  "Salience (0=Low, 1=High)"
        rt=   "rt: Stacked Response Time across Conditions";
Drop old multivariate outcomes;
  DROP rt11--rt22;
*
RUN;

* Create additional variables -- cannot be done right after stacking code;
DATA work.Example4a; SET work.Example4a;
* Log RT to fit log-normal conditional distribution;
  logRT=LOG(RT);
* Create piecewise effects of age;
       IF old=0 THEN yrs65=0;
  ELSE IF old=1 THEN yrs65=age-65;
* Label new variables;
  LABEL logrt= "logRT: Natural Log of Response Time"
        yrs65= "yrs65: Age in Older Adult Group (0=65)";
RUN; * Sort by condition (needed for later);
PROC SORT DATA=work.Example4a; BY condition PersonID; RUN;
```

Empty Multivariate Model Predicting Log RT: Predict the RT in condition *c* for person *i*:

$$\widehat{RT}_{ic} = \beta_{00} + \beta_{01}(Mean_{ic}) + \beta_{02}(Sal_{ic}) + \beta_{03}(Mean_{ic})(Sal_{ic})$$

Although this model doesn't look empty, it is—each of the four outcomes has its own mean and there are no other predictors (yet). Outcome means are thus created by:

	Low Salience	High Salience
Low Meaning	β_{00}	$\beta_{00} + \beta_{02}$
High Meaning	$\beta_{00} + \beta_{01}$	$\beta_{00} + \beta_{01} + \beta_{02} + \beta_{03}$

Let's start with the "**answer key**" model for the variance: An **unstructured R matrix** in which all variances and covariances across the four outcomes are estimated separately (i.e., "multivariate" ANOVA except estimated using REML instead of least squares to avoid listwise deletion of persons with incomplete outcomes):

```
display as result "STATA Empty Multivariate Model: Unstructured R Matrix"
mixed logrt c.mean#c.sal, ///
          || personid: , noconstant variance reml ///
          dfmethod(satterthwaite) dftable(pvalue) ///
          residuals(unstructured,t(condition)),
       estat wcorrelation, covariance,
       estat wcorrelation,
       estimates store UN
```

STATA: || personid: . noconstant identifies nesting structure of conditions within persons without adding any person-level additional variances

estat wcorrelation, covariance → R matrix
estat wcorrelation → RCORR matrix

```
TITLE "SAS Empty Multivariate Model: Unstructured R Matrix";
PROC MIXED DATA=work.Example4a COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
     CLASS PersonID condition;
     MODEL logRT = mean|sal@2 / DDFM=Satterthwaite;
     REPEATED condition / R RCORR TYPE=UN SUBJECT=PersonID;
RUN; TITLE;
```

SAS: R and RCORR to show in output

SAS Output from Unstructured R Matrix model:

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	788.40028446	
1	1	**336.55960475**	0.00000000

For your homework using SAS, get your −2LL value from this table to get two digits after the decimal.

Estimated R Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	**0.1366**	0.1296	0.1205	0.1254
2	0.1296	**0.2369**	0.1676	0.1652
3	0.1205	0.1676	**0.2291**	0.1673
4	0.1254	0.1652	0.1673	**0.2059**

This **R matrix** holds the variances and covariances across conditions. Given complete data (not required), it will exactly match those in original data.

Do the **variances** appear to differ across conditions?

Estimated R Correlation Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	1.0000	0.7207	0.6814	0.7479
2	**0.7207**	1.0000	0.7194	0.7481
3	**0.6814**	**0.7194**	1.0000	0.7705
4	**0.7479**	**0.7481**	**0.7705**	1.0000

This **RCORR matrix** holds the correlations across conditions. Given complete data (not required), it will exactly match those in the original data.

Do the **correlations** appear to differ across conditions?

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	PersonID	0.1366	0.01551	8.80	<.0001
UN(2,1)	PersonID	0.1296	0.01781	7.28	<.0001
UN(2,2)	PersonID	0.2369	0.02692	8.80	<.0001
UN(3,1)	PersonID	0.1205	0.01719	7.01	<.0001
UN(3,2)	PersonID	0.1676	0.02305	7.27	<.0001
UN(3,3)	PersonID	0.2291	0.02602	8.80	<.0001
UN(4,1)	PersonID	0.1254	0.01682	7.46	<.0001
UN(4,2)	PersonID	0.1652	0.02216	7.46	<.0001
UN(4,3)	PersonID	0.1673	0.02202	7.60	<.0001
UN(4,4)	PersonID	0.2059	0.02339	8.80	<.0001

This "CovParms" table lists each parameter estimated as part of the model for the variance (i.e., unique entry in the **R** matrix here).

"UN(r,c)" labels parameters from the unstructured **R** matrix as (rows, columns).

These Wald test *p*-values should not be used!

Fit Statistics

-2 Res Log Likelihood	**336.56**
AIC (Smaller is Better)	356.6
AICC (Smaller is Better)	356.9
BIC (Smaller is Better)	387.1

This is the sum of the individual log-likelihoods multiplied by −2. It is the best possible fit for the model for the variance.

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
9	451.84	<.0001

This "null model" LRT gives the test of the current model for the variance (UN) vs. just a single homogeneous residual variance with no covariances (VC). It's too general to be helpful right now.

Now let's see if we could use a simpler model: Compound Symmetry Heterogeneous, in which all variances differ (so covariances still differ) but all correlations are held equal to "CSH" (not in STATA):

```
TITLE "SAS Empty Multivariate Model: Compound Symmetry Heterogeneous R Matrix";
PROC MIXED DATA=work.Example4a COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
     CLASS PersonID condition;
     MODEL logRT = mean|sal@2 / SOLUTION DDFM=Satterthwaite;
     REPEATED condition / R RCORR TYPE=CSH SUBJECT=PersonID; RUN; TITLE;
```

SAS Output from Compound Symmetry Heterogeneous R Matrix model:

Estimated R Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	**0.1389**	0.1328	0.1310	0.1220
2	0.1328	**0.2375**	0.1713	0.1596
3	0.1310	0.1713	**0.2310**	0.1574
4	0.1220	0.1596	0.1574	**0.2004**

This **R matrix** still allows the residual variances to differ by condition, but the covariances are constrained—as the CSH common correlation multiplied by the SD for each condition.

Estimated R Correlation Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	1.0000	0.7315	0.7315	0.7315
2	**0.7315**	1.0000	0.7315	0.7315
3	**0.7315**	**0.7315**	1.0000	0.7315
4	**0.7315**	**0.7315**	**0.7315**	1.0000

This **RCORR matrix** now predicts all residual correlations to be the CSH correlation = 0.7315.

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
Var(1)	PersonID	0.1389	0.01573	8.83	<.0001
Var(2)	PersonID	0.2375	0.02679	8.87	<.0001
Var(3)	PersonID	0.2310	0.02610	8.85	<.0001
Var(4)	PersonID	0.2004	0.02244	8.93	<.0001
CSH	PersonID	0.7315	0.02813	26.01	<.0001

This CSH model has five unique parameters—4 residual variances (labeled "Var(n)") and one CSH common correlation (as seen directly in RCORR above).

Fit Statistics

-2 Res Log Likelihood	**343.23**
AIC (Smaller is Better)	353.2
AICC (Smaller is Better)	353.3
BIC (Smaller is Better)	368.5

Does this CSH model with 5 parameters fit worse than the UN model with 10 parameters (1 for each possible variance and covariance; −2LL = 336.56)?

−2ΔLL (5) = 343.23 – 336.56 = 6.67, *p* = .246, so CSH is not worse than UN

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
4	445.17	<.0001

This "null model" LRT gives the test of the current model for the variance (CSH) vs. just a single homogeneous residual variance with no covariances (VC). It's still too general to be helpful right now.

Now let's see if we can use an even simpler model: Compound Symmetry, in which all variances are predicted to be equal and all covariances are predicted to be equal, too (i.e., "Univariate" ANOVA):

```
display as result "STATA Empty Multivariate Model: Compound Symmetry R Matrix"
mixed logrt c.mean##c.sal, || personid: , noconstant variance reml ///
         dfmethod(satterthwaite) dftable(pvalue) residuals(exchangeable,t(condition)),
      estat wcorrelation, covariance,
      estat wcorrelation,
      estimates store CS
      lrtest UN CS
```

STATA: estimates store saves results, lrtest then requests likelihood ratio test against UN model

```
TITLE "SAS Empty Multivariate Model: Compound Symmetry R Matrix";
PROC MIXED DATA=work.Example4a COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
     CLASS PersonID condition;
     MODEL logRT = mean|sal@2 / SOLUTION DDFM=Satterthwaite;
     REPEATED condition / R RCORR TYPE=CS SUBJECT=PersonID; RUN; TITLE;
```

SAS Output from Compound Symmetry R Matrix model:

Estimated R Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	0.2021	0.1460	0.1460	0.1460
2	0.1460	0.2021	0.1460	0.1460
3	0.1460	0.1460	0.2021	0.1460
4	0.1460	0.1460	0.1460	0.2021

This **R matrix** now predicts the residual variance to be 0.2021 regardless of condition. Part of it (0.1460) is due to mean RT differences across persons (as CS), and the rest (0.2021 – 0.1460 = 0.056) is from within-condition residual variation.

Estimated R Correlation Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	1.0000	0.7221	0.7221	0.7221
2	0.7221	1.0000	0.7221	0.7221
3	0.7221	0.7221	1.0000	0.7221
4	0.7221	0.7221	0.7221	1.0000

This **RCORR matrix** now predicts the residual correlation to be 0.7221 regardless of condition.

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
CS	PersonID	0.1460	0.01820	8.02	<.0001
Residual		0.05617	0.003684	15.25	<.0001

This table gives the separately estimated parameters that create the R matrix pattern. Do NOT use these *p*-values!

Fit Statistics

-2 Res Log Likelihood	**371.59**
AIC (smaller is better)	375.6
AICC (smaller is better)	375.6
BIC (smaller is better)	381.7

Does this CS model with only 2 parameters fit worse than the CSH model with 5 parameters (with 4 separate variances instead; −2LL = 343.23)?

−2ΔLL (3) = 371.59 – 343.23 = 9.19, $p < .001$, so CS fits worse than CSH
−2ΔLL (8) = 371.59 – 336.56 = 35.03, $p < .001$, so CS fits worse than UN

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
1	416.81	<.0001

The null model LRT gives the −2LL difference against TYPE=VC (no covariance across outcomes).

Let's examine differences in the fixed effects solution across the UN, CSH, and CS R matrices:

Solution for Fixed Effects from Unstructured R

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	2.4195	**0.02959**	155	81.77	<.0001	Beta00
mean	-0.1782	0.02826	155	-6.31	<.0001	Beta01
sal	-0.3478	0.02706	155	-12.85	<.0001	Beta02
mean*sal	0.07564	0.03832	155	1.97	0.0502	Beta03

Btw, the Satterthwaite denominator DF method adjusts for differences in sample size and variance across repeated measures for any R matrix type other than UN (in which case it just uses DF based on the number of persons, $N = 156$ here).

Solution for Fixed Effects from Compound Symmetry Heterogeneous R

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	2.4195	**0.02984**	156	81.09	<.0001	Beta00
mean	-0.1782	0.02629	277	-6.78	<.0001	Beta01
sal	-0.3478	0.02664	272	-13.06	<.0001	Beta02
mean*sal	0.07564	0.03863	390	1.96	0.0510	Beta03

As further evidence that CSH is sufficient relative to the more complex UN, their fixed effects SEs are very similar.

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	2.4195	**0.03600**	242	67.22	<.0001	Beta00
mean	-0.1782	0.02683	465	-6.64	<.0001	Beta01
sal	-0.3478	0.02683	465	-12.96	<.0001	Beta02
mean*sal	0.07564	0.03795	465	1.99	0.0468	Beta03

Note that when all variances are constrained equal in the CS model, the intercept SE (for the reference outcome of low/low) is higher than when each outcome gets its own variance.

Given its significantly worse fit than CSH or UN, CS should not be used. CSH could be used given that it fit not worse than UN, but CSH is not available in STATA. So I will proceed using an **unstructured R matrix**.

Here is the new predictive adding age-related fixed main effects and interaction effects:

$$\widehat{RT}_{ic} = \beta_{00} + \beta_{01}(Mean_{ic}) + \beta_{02}(Sal_{ic}) + \beta_{03}(Mean_{ic})(Sal_{ic})$$
$$+\beta_{10}(Old_i) + \beta_{11}(Mean_{ic})(Old_i) + \beta_{12}(Sal_{ic})(Old_i) + \beta_{13}(Mean_{ic})(Sal_{ic})(Old_i)$$
$$+\beta_{20}(Yrs65_i) + \beta_{21}(Mean_{ic})(Yrs65_i) + \beta_{22}(Sal_{ic})(Yrs65_i) + \beta_{23}(Mean_{ic})(Sal_{ic})(Yrs65_i)$$

Given these fixed effects, I want to compare a more complex model allowing separate unstructured R matrices by age group to a less complex model constraining the R matrix to be the same across groups:

```
display as result "STATA Predictive Multivariate Model: Add Age Group and Years over 65"
display as result "Different Unstructured R Matrix per Age Group"
mixed logrt c.mean##c.sal##c.old c.mean##c.sal##c.yrs65, ///
        || personid: , noconstant variance reml dfmethod(satterthwaite) dftable(pvalue) ///
        residuals(unstructured,t(condition) by(old)),  // by: get R separate by Old
      estat ic, n(156),
      estat wcorrelation, covariance at(personid=1)   // R for Young (first)
      estat wcorrelation, covariance at(personid=274) // R for Old (last)
      estat wcorrelation, at(personid=1)              // RCORR for Young (first)
      estat wcorrelation, at(personid=274)            // RCORR for Old (last)

TITLE1 "SAS Predictive Multivariate Model: Add Age Group and Years over 65";
TITLE2 "Different Unstructured R Matrix per Age Group";
PROC MIXED DATA=work.Example4a COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
     CLASS PersonID condition;
     MODEL logRT = mean|sal|old@3 mean|sal|yrs65@3 / SOLUTION DDFM=Satterthwaite;
     * GROUP=old gets separate R by age group (print R, CORR for first, last person);
     REPEATED condition / R=1,156 RCORR=1,156 TYPE=UN SUBJECT=PersonID GROUP=old;
RUN; TITLE1; TITLE2;
```

SAS Output from Separate Unstructured R Matrix per Age Group model (truncated):

Estimated R Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	**0.03936**	0.005212	-0.00352	0.006988
2	0.005212	**0.07176**	0.005092	0.009808
3	-0.00352	0.005092	**0.07099**	0.01472
4	0.006988	0.009808	0.01472	**0.06227**

Estimated R Correlation Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	1.0000	0.09807	-0.06660	0.1412
2	0.09807	1.0000	0.07135	0.1467
3	-0.06660	0.07135	1.0000	0.2214
4	0.1412	0.1467	0.2214	1.0000

Estimated R Matrix for PersonID 274

Row	Col1	Col2	Col3	Col4
1	**0.07165**	0.02565	0.01615	0.02708
2	0.02565	**0.08480**	0.01326	0.01857
3	0.01615	0.01326	**0.07117**	0.01863
4	0.02708	0.01857	0.01863	**0.06275**

Estimated R Correlation Matrix for PersonID 274

Row	Col1	Col2	Col3	Col4
1	1.0000	0.3291	0.2261	0.4039
2	0.3291	1.0000	0.1707	0.2546
3	0.2261	0.1707	1.0000	0.2787
4	0.4039	0.2546	0.2787	1.0000

Fit Statistics

-2 Res Log Likelihood	83.37
AIC (Smaller is Better)	123.4
AICC (Smaller is Better)	124.8
BIC (Smaller is Better)	184.4

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	2.1926	0.02014	96	108.85	<.0001	Beta00
mean	-0.2592	0.03479	96	-7.45	<.0001	Beta01
sal	-0.4272	0.03222	96	-13.26	<.0001	Beta02
mean*sal	0.1706	0.04720	96	3.61	0.0005	Beta03
old	0.5149	0.08189	64.4	6.29	<.0001	Beta10
mean*old	0.1490	0.1045	71.4	1.42	0.1585	Beta11
sal*old	0.03475	0.1014	70	0.34	0.7329	Beta12
mean*sal*old	-0.1035	0.1448	70.8	-0.71	0.4771	Beta13
yrs65	0.007829	0.006564	57	1.19	0.2380	Beta20
mean*yrs65	0.006010	0.008152	57	0.74	0.4640	Beta21
sal*yrs65	0.01611	0.007952	57	2.03	0.0475	Beta22
mean*sal*yrs65	-0.01358	0.01132	57	-1.20	0.2353	Beta23

```
display as result "STATA Predictive Multivariate Model: Add Age Group and Years over 65"
display as result "Same Unstructured R Matrix across Age Groups"
mixed logrt c.mean##c.sal##c.old c.mean##c.sal##c.yrs65, ///
        || personid: , noconstant variance reml dfmethod(satterthwaite) dftable(pvalue) ///
        residuals(unstructured,t(condition)),
      estat wcorrelation, covariance
      estat wcorrelation

TITLE1 "SAS Predictive Multivariate Model: Add Age Group and Years over 65";
TITLE2 "Same Unstructured R Matrix across Age Groups";
PROC MIXED DATA=work.Example4a COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
     CLASS PersonID condition;
     MODEL logRT = mean|sal|old@3 mean|sal|yrs65@3 / SOLUTION DDFM=Satterthwaite;
     REPEATED condition / R RCORR TYPE=UN SUBJECT=PersonID;
RUN; TITLE1; TITLE2;
```

SAS Output from Same Unstructured R Matrix across Age Groups model (truncated):

Estimated R Matrix for PersonID 1 (same for all)

Row	Col1	Col2	Col3	Col4
1	0.05139	0.01283	0.003807	0.01447
2	0.01283	0.07662	0.008135	0.01307
3	0.003807	0.008135	0.07105	0.01618
4	0.01447	0.01307	0.01618	0.06245

Estimated R Correlation Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	1.0000	0.2044	0.06300	0.2555
2	0.2044	1.0000	0.1103	0.1890
3	0.06300	0.1103	1.0000	0.2428
4	0.2555	0.1890	0.2428	1.0000

Fit Statistics

-2 Res Log Likelihood	94.84
AIC (Smaller is Better)	114.8
AICC (Smaller is Better)	115.2
BIC (Smaller is Better)	145.3

Does this same-UN model with 10 parameters fit worse than the age-UN model with 20 parameters (keeping all fixed effects the same)?

$-2\Delta LL$ (10) = 94.84 – 83.37 = 11.47, p = .322, so same-UN is not worse

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	2.1926	0.02302	153	95.26	<.0001	Beta00
mean	-0.2592	0.03441	153	-7.53	<.0001	Beta01
sal	-0.4272	0.03248	153	-13.15	<.0001	Beta02
mean*sal	0.1706	0.04708	153	3.62	0.0004	Beta03
old	0.5149	0.07105	153	7.25	<.0001	Beta10
mean*old	0.1490	0.1062	153	1.40	0.1628	Beta11
sal*old	0.03475	0.1003	153	0.35	0.7294	Beta12
mean*sal*old	-0.1035	0.1453	153	-0.71	0.4774	Beta13
yrs65	0.007829	0.005559	153	1.41	0.1611	Beta20
mean*yrs65	0.006010	0.008310	153	0.72	0.4706	Beta21
sal*yrs65	0.01611	0.007846	153	2.05	0.0417	Beta22
mean*sal*yrs65	-0.01358	0.01137	153	-1.19	0.2342	Beta23 → NS either way

After removing yrs65*mean*sal, the two-way interactions of yrs65*mean and yrs65*sal were still not significant, so those were removed, leaving only the significant main effect of yrs65.

Here is the reduced predictive model (in which the highest-order interaction is significant):

$$\widehat{RT}_{ic} = \beta_{00} + \beta_{01}(Mean_{ic}) + \beta_{02}(Sal_{ic}) + \beta_{03}(Mean_{ic})(Sal_{ic})$$
$$+\beta_{10}(Old_i) + \beta_{11}(Mean_{ic})(Old_i) + \beta_{12}(Sal_{ic})(Old_i) + \beta_{13}(Mean_{ic})(Sal_{ic})(Old_i) + \beta_{20}(Yrs65_i)$$

```
display as result "STATA Reduced Predictive Multivariate Model: Main Effect Years over 65"
display as result "Age-Constrained Unstructured R Matrix"
mixed logrt c.mean##c.sal##c.old c.yrs65, ///
        || personid: , noconstant variance reml dfmethod(satterthwaite) dftable(pvalue) ///
        residuals(exchangeable,t(condition)),
      estat wcorrelation, covariance,
      estat wcorrelation,
      predict pred, xb // Add column pred of predicted outcomes to data
// Simple slopes for meaning, by salience and age
   lincom c.mean*1 + c.mean#c.sal*0 + c.mean#c.old*0 + c.mean#c.sal#c.old*0, small // LvH Mean:LS Y
   lincom c.mean*1 + c.mean#c.sal*1 + c.mean#c.old*0 + c.mean#c.sal#c.old*0, small // LvH Mean:HS Y
   lincom c.mean*1 + c.mean#c.sal*0 + c.mean#c.old*1 + c.mean#c.sal#c.old*0, small // LvH Mean:LS O
   lincom c.mean*1 + c.mean#c.sal*1 + c.mean#c.old*1 + c.mean#c.sal#c.old*1, small // LvH Mean:HS O
// Simple slopes for salience, by meaning and age
   lincom c.sal*1 + c.mean#c.sal*0 + c.sal#c.old*0 + c.mean#c.sal#c.old*0, small // LvH Sal:LM Y
   lincom c.sal*1 + c.mean#c.sal*1 + c.sal#c.old*0 + c.mean#c.sal#c.old*0, small // LvH Sal:HM Y
   lincom c.sal*1 + c.mean#c.sal*0 + c.sal#c.old*1 + c.mean#c.sal#c.old*0, small // LvH Sal:LM O
   lincom c.sal*1 + c.mean#c.sal*1 + c.sal#c.old*1 + c.mean#c.sal#c.old*1, small // LvH Sal:HM O
// Simple meaning*salience interactions, by age
   lincom c.mean#c.sal*1 + c.mean#c.sal#c.old*0, small // Mean*Sal: Y
   lincom c.mean#c.sal*1 + c.mean#c.sal#c.old*1, small // Mean*Sal: O
// Predicted means per condition
   margins, at(c.mean=(0(1)1) c.sal=(0(1)1) c.old=(0(1)1) c.yrs65=0)
   marginsplot // Make plot of requested margins
// Get correlation of actual and predicted outcomes to form R2
   pwcorr logrt pred if condition==11, sig
   pwcorr logrt pred if condition==12, sig
   pwcorr logrt pred if condition==21, sig
   pwcorr logrt pred if condition==22, sig
```

“small” means use same denominator DF method as for fixed effects above

```
TITLE1 "SAS Reduced Predictive Multivariate Model: Main Effect Years over 65";
TITLE2 "Age-Constrained Unstructured R Matrix";
PROC MIXED DATA=work.Example4a COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
     CLASS PersonID condition;          * OUTPM saves dataset of predicted outcomes;
     MODEL logRT = mean|sal|old@3 yrs65   /* RESIDUAL adds plots of residuals */
                         / RESIDUAL SOLUTION DDFM=Satterthwaite OUTPM=work.PredFinal;
     REPEATED condition / R RCORR TYPE=UN SUBJECT=PersonID;
* Simple slopes for meaning, by salience and age;
     ESTIMATE "LvsH Mean: Sal=Low,  Young"   mean 1 mean*sal 0 mean*old 0 mean*sal*old 0;
     ESTIMATE "LvsH Mean: Sal=High, Young"   mean 1 mean*sal 1 mean*old 0 mean*sal*old 0;
     ESTIMATE "LvsH Mean: Sal=Low,  Old"     mean 1 mean*sal 0 mean*old 1 mean*sal*old 0;
     ESTIMATE "LvsH Mean: Sal=High, Old"     mean 1 mean*sal 1 mean*old 1 mean*sal*old 1;
* Simple slopes for salience, by meaning and age;
     ESTIMATE "LvsH Sal: Mean=Low,  Young"   sal 1 mean*sal 0 sal*old 0 mean*sal*old 0;
     ESTIMATE "LvsH Sal: Mean=High, Young"   sal 1 mean*sal 1 sal*old 0 mean*sal*old 0;
     ESTIMATE "LvsH Sal: Mean=Low,  Old"     sal 1 mean*sal 0 sal*old 1 mean*sal*old 0;
     ESTIMATE "LvsH Sal: Mean=High, Old"     sal 1 mean*sal 1 sal*old 1 mean*sal*old 1;
* Simple mean*sal interactions, by age;
     ESTIMATE "Mean*Sal: Young" mean*sal 1 mean*sal*old 0;
     ESTIMATE "Mean*Sal: Old"   mean*sal 1 mean*sal*old 1;
* Predicted means per condition (years65=0);
     ESTIMATE "Int: Young, Low Mean, Low Sal"    intercept 1 mean 0 sal 0 mean*sal 0 old 0
                                                  mean*old 0 sal*old 0 mean*sal*old 0;
     ESTIMATE "Int: Young, Low Mean, High Sal"   intercept 1 mean 0 sal 1 mean*sal 0 old 0
                                                  mean*old 0 sal*old 0 mean*sal*old 0;
     ESTIMATE "Int: Young, High Mean, Low Sal"   intercept 1 mean 1 sal 0 mean*sal 0 old 0
                                                  mean*old 0 sal*old 0 mean*sal*old 0;
     ESTIMATE "Int: Young, High Mean, High Sal"  intercept 1 mean 1 sal 1 mean*sal 1 old 0
                                                  mean*old 0 sal*old 0 mean*sal*old 0;
     ESTIMATE "Int: Old, Low Mean, Low Sal"      intercept 1 mean 0 sal 0 mean*sal 0 old 1
                                                  mean*old 0 sal*old 0 mean*sal*old 0;
     ESTIMATE "Int: Old, Low Mean, High Sal"     intercept 1 mean 0 sal 1 mean*sal 0 old 1
                                                  mean*old 0 sal*old 1 mean*sal*old 0;
     ESTIMATE "Int: Old, High Mean, Low Sal"     intercept 1 mean 1 sal 0 mean*sal 0 old 1
                                                  mean*old 1 sal*old 0 mean*sal*old 0;
     ESTIMATE "Int: Old, High Mean, High Sal"    intercept 1 mean 1 sal 1 mean*sal 1 old 1
                                                  mean*old 1 sal*old 1 mean*sal*old 1;
```

```
    ODS OUTPUT Estimates=work.EstSave; * Save ESTIMATEs for plotting;
RUN; TITLE;
```

SAS Output for Final Model (truncated):

Estimated R Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	0.05166	0.01223	0.003968	0.01439
2	0.01223	0.07729	0.008197	0.01337
3	0.003968	0.008197	0.07075	0.01622
4	0.01439	0.01337	0.01622	0.06225

Estimated R Correlation Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	1.0000	0.1935	0.06563	0.2538
2	0.1935	1.0000	0.1108	0.1927
3	0.06563	0.1108	1.0000	0.2444
4	0.2538	0.1927	0.2444	1.0000

Fit Statistics

-2 Res Log Likelihood	**75.10**
AIC (Smaller is Better)	95.1
AICC (Smaller is Better)	95.5
BIC (Smaller is Better)	125.6

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	2.1926	0.02308	153	95.01	<.0001	Beta00
mean	-0.2592	0.03435	154	-7.55	<.0001	Beta01
sal	-0.4272	0.03282	154	-13.02	<.0001	Beta02
mean*sal	0.1706	0.04714	154	3.62	0.0004	Beta03
old	0.4451	0.05595	207	7.96	<.0001	Beta10
mean*old	0.2143	0.05586	154	3.84	0.0002	Beta11
sal*old	0.2098	0.05337	154	3.93	0.0001	Beta12
mean*sal*old	-0.2510	0.07665	154	-3.27	0.0013	Beta13
yrs65	0.01425	0.003820	153	3.73	0.0003	Beta20

How years of age adjusts the intercept in older adults is same for all conditions

Interpret the fixed effects:

β_{00} Intercept = logRT when mean=low, sal=low, old=YA

β_{01} Meaning = low vs. high meaning when sal=low and old=YA

β_{02} Salience = low vs. high salience when mean=low and old=YA

β_{03} Meaning*Salience = reduction in low vs. high meaningful effect when sal=high (and old=YA)

reduction in low vs. high salience effect when mean=high (and old=YA)

β_{10} Old = YA vs. OA when mean=low and sal=low

β_{11} Old*Meaning = reduction in low vs. high meaningful effect when old=OA (and sal=low)

β_{12} Old*Salience = reduction in low vs. high salience effect when old=OA (and mean=low)

β_{13} Old*Meaning*Salience = the extent that high salience reduces the meaning effect in YA is reduced in OA

β_{20} Years over 65 = for every year older than 65 in OA, logRT is higher by 0.01425

```
* Get R2 per outcome condition (prediction by old and yrs65);
PROC SORT DATA=work.PredFinal; BY condition PersonID; RUN;
PROC CORR NOSIMPLE DATA=work.PredFinal; BY condition; VAR logrt; WITH pred; RUN;
```

condition: Index for Outcome (M/S)=11	logRT	condition: Index for Outcome (M/S)=12	logRT
Pred	0.79082 → R^2 = .6254	Pred	0.82257 → R^2 = .6766
Predicted Mean	<.0001	Predicted Mean	<.0001
condition: Index for Outcome (M/S)=21	**logRT**	**condition: Index for Outcome (M/S)=22**	**logRT**
Pred	0.83296 → R^2 = .6938	Pred	0.83692 → R^2 = .7004
Predicted Mean	<.0001	Predicted Mean	<.0001

```
* Calculate effect sizes from estimates;
DATA work.EstEffect; SET work.EstSave;
     WHERE INDEX(Label, "Int:")=0; * Exclude intercepts;
     r=tvalue/SQRT((tvalue*tvalue)+DF);
     d=2*tvalue/SQRT(DF); RUN;
PROC PRINT NOOBS DATA=work.EstEffect; RUN;
```

Estimates → Will be Table X in results

Label	Estimate	StdErr	DF	tValue	Probt	r	d
LvsH Mean: Sal=Low, Young	-0.2592	0.03435	154	-7.55	<.0001	-0.51956	-1.21617
LvsH Mean: Sal=High, Young	-0.08866	0.03410	154	-2.60	0.0102	-0.20506	-0.41902
LvsH Mean: Sal=Low, Old	-0.04498	0.04405	154	-1.02	0.3088	-0.08201	-0.16457
LvsH Mean: Sal=High, Old	-0.1254	0.04372	154	-2.87	0.0047	-0.22518	-0.46224
LvsH Sal: Mean=Low, Young	-0.4272	0.03282	154	-13.02	<.0001	-0.72373	-2.09752
LvsH Sal: Mean=High, Young	-0.2566	0.03220	154	-7.97	<.0001	-0.54035	-1.28436
LvsH Sal: Mean=Low, Old	-0.2174	0.04208	154	-5.17	<.0001	-0.38430	-0.83253
LvsH Sal: Mean=High, Old	-0.2978	0.04128	154	-7.21	<.0001	-0.50257	-1.16263
Mean*Sal: Young	0.1706	0.04714	154	3.62	0.0004	0.27992	0.58316
Mean*Sal: Old	-0.08043	0.06044	154	-1.33	0.1853	-0.10661	-0.21445

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
(all simple effects and simple effect differences are given above)					
Int: Young, Low Mean, Low Sal	2.1926	0.02308	153	95.01	<.0001
Int: Young, Low Mean, High Sal	1.7654	0.02823	153	62.54	<.0001
Int: Young, High Mean, Low Sal	1.9334	0.02701	154	71.59	<.0001
Int: Young, High Mean, High Sal	1.6768	0.02533	154	66.19	<.0001
Int: Old, Low Mean, Low Sal	2.6377	0.05097	200	51.75	<.0001
Int: Old, Low Mean, High Sal	2.4203	0.05506	231	43.95	<.0001
Int: Old, High Mean, Low Sal	2.5927	0.05405	225	47.97	<.0001
Int: Old, High Mean, High Sal	2.2949	0.05270	215	43.55	<.0001

```
* Subset and re-arrange estimates for plotting;
DATA work.EstPlot; SET work.EstSave;
     WHERE INDEX(Label, "Int:")>0; * Only include intercepts;
     IF INDEX(Label,"Old")>0 THEN old=1; ELSE old=0;
     IF INDEX(Label,"High Mean")>0 THEN mean=1; ELSE mean=0;
     IF INDEX(Label,"High Sal")>0 THEN sal=1; ELSE sal=0; RUN;

* Add value labels for use in plot below;
PROC FORMAT;
     VALUE Fold 0="Younger" 1="Older";
     VALUE Fcond 0="Low" 1="High"; RUN;
TITLE "Figure for Three-Way Interaction";
PROC SGPANEL DATA=work.EstPlot;
     PANELBY old / ROWS=1 COLUMNS=2 NOVARNAME;
     SERIES x=mean y=Estimate  / GROUP=sal;
     COLAXIS LABEL="Meaningfulness to Driving" VALUES=(0 TO 1);
     ROWAXIS LABEL="Predicted Log RT" VALUES=(1.5 TO 3.0 BY 0.1);
     FORMAT old Fold. mean Fcond. sal Fcond.;
     LABEL sal="Visual Salience"; RUN;

TITLE "SAS Separate by Outcome in GLIMMIX to check Pearson chi-square / DF";
PROC GLIMMIX DATA=work.Example4a NOCLPRINT NAMELEN=100 METHOD=RSPL PLOTS=(ALL);
     BY condition;   * Separate by outcome, RSPL=REML, add PLOTS for residuals;
     MODEL logRT = old yrs65 / SOLUTION LINK=IDENTITY DIST=NORMAL DDFM=Satterth;
RUN; TITLE;

display as result "STATA Separate by Outcome in GLM to check Pearson chi-square / DF"
bysort condition: glm logrt c.old c.yrs65, link(identity) family(gaussian)
```



This plot from the RESIDUAL option in SAS MIXED suggests normal residuals for Log-RT fit well.
The Pearson χ^2/DF ranged from 0.05–0.08 across the condition outcomes (so no overdispersion for sure).

Results section using SAS output (skipping CSH since we didn't use it):

We examined the extent to which response time (RT) to detect changes in driving scenes could be predicted by two repeated measures factors: whether the changes were of low or high meaningfulness to driving, and whether the changes were of low or high visual salience. We also included a between-subjects predictor for age group (younger or older adult), along with a covariate of years over age 65 in the older adults. Given RT's positive skewness, we predicted log-transformed RT instead (i.e., such that the model residuals were assumed to follow a log-normal distribution instead of a normal distribution). All models were estimated in SAS MIXED using residual maximum likelihood (REML), which is equivalent to ordinary least squares given complete outcomes per person. The Satterthwaite method was used to estimate denominator degrees of freedom, and the fit of alternative models for the pattern of variance and covariance across the four condition outcomes was compared using likelihood ratio tests (i.e., by treating the difference in −2LL between models as a χ^2 with degrees of freedom equal to the difference in the number of parameters). ESTIMATE statements were used to estimate simple slopes and simple slope differences as linear combinations of the model fixed effects. Effect sizes are given as model R^2 per condition, as well as in standardized mean difference units (d) and correlation units (r) calculated from the Wald test statistics for the corresponding fixed effects (or linear combinations thereof).

We first examined the pattern of RT variance and covariance across the four condition outcomes (low/high meaning crossed with low/high salience) while allowing separate means by condition. Relative to an unstructured model (i.e., in which all four variances and all six pairwise covariances were estimated separately, a multivariate approach), a compound symmetry model (i.e., in which all variances were constrained to be equal and all covariances were constrained to be equal, a univariate approach) fit significantly worse, $-2\Delta LL(8) = 35.03$, $p < .001$. Consequently, we retained the unstructured model and added all possible interactions of both age group and years over age 65 with meaning, salience, and their interaction. We then tested for heterogeneity of variance by age group of the variances and covariances across the four condition outcomes, but no evidence was found, $-2\Delta LL(10) = 11.47$, $p = .322$. In addition, while the three-way interaction of meaning by salience by age group was significant, only the main effect of years over 65 was significant. Consequently, we retained all significant fixed effects (and their lower-order terms) and a common matrix of variances and covariances across the four outcome conditions. Results are described below, provided in Table X, and shown in Figure X.

Slower response times in the older age group and with additional years over age 65 accounted for 63–70% of the variance within conditions. In the younger adults, RT was significantly faster for changes of high than low meaning in both low and high salience, although this effect of meaning was significantly greater for changes of low than high salience. In the older adults, RT was significantly faster to changes of high than low meaning, but only for changes of high salience (with no significant difference for changes of low salience); the meaning by salience interaction was also not significant.